
Chapter 4 - Data Buffering
Chapter Objectives

After completing this chapter, you will be able to:

· Use Private Data Sessions 

· Use Data Buffering

· Know the different flavors of data buffering

· Use TableUpdate()

· Use Transactions

· Know how to share data between forms when necessary

Using Private Data Sessions

Visual FoxPro Forms have either a default or private data session.  

If the data session is private it has its own set of tables.  Even if other forms are using the same tables, they will not interfere with each other.  Moving the record pointer in one form will not move it in another.  

You will usually use private data sessions.

If the data session is default it will use whatever data session is open in the calling form.
Use default data session ONLY when you need to share data between a main form that is calling a modal dialog form.  Even then, it is cleaner to just pass data as parameters and return values in most cases.

It is generally a bad idea to use default data sessions.
Two or more open forms can and will interfere with data in other forms.
The interference will include changing the current work area, closing tables you need open, etc.

Open form frmPlain and use the property sheet and change the form to private data session:

Using Data Buffering

Data buffering puts your table in a “just kidding” mode.
You put your table in one of VFP’s buffering modes, and you can edit records without actually changing the table on disk.  If your user decides to save, then you can tell VFP you were serious in one line of code.
In the old days when you changed a record, it went straight to disk.
To provide save/cancel functionality, we would “scatter” fields from the disk into memory and let the user edit memory variables.  If the user wanted to save we would “gather” the memory variables into the database record on disk.  
If you wanted to let your user edit multiple records and still have save/cancel functionality, you would have to create a temp table on disk with the records you needed, and then write some code to loop through your temp table and transfer the values to the real table when the user saved.
Ugh.

VFP makes it so much easier!

1. Right Click on the form and select "Data Environment"

2. Right Click on the person table and select "Properties"

3. Select "BufferModeOverride" and set it to 5 – Optimistic Table Buffering

4. Run the form again and change some values.  
5. Then close the form.

6. Run the form again – the values did not change!

The values did not change because we had the table buffered.
We told VFP we were just kidding and we never told it we were serious enough to save the changes to disk.
Buffer Modes

Row Buffering is not that different than what you get natively.  
Row Buffering is used sometimes temporarily, for instance to add indexes to a buffered table or view, you must NOT have table buffering on.

Table Buffering is what you will really use most of the time in your apps.
None of the users changes is saved until you explicitly save them with TABLEUPDATE() 
If your user decides to cancel, you can lose all changes by issuing a TABLEREVERT()
VFP has done an outstanding job with this functionality.
It works the same way on native FoxPro tables, FoxPro local views, and remote views against servers like SQL Server.

Pessimistic means that VFP will lock all your edited records.
The records are locked until you issue a TABLEUPDATE() or TABLEREVERT()
This only works on VFP tables.
It will not work for VFP views against VFP tables.
It will not work for VFP views against server data like SQL Server.
SET MULTILOCKS ON for this to work properly.
Optimistic means that VFP will not lock edited records.
This works against VFP tables, VFP local views, and VFP remote views against server tables like SQL Server.

Why can’t server databases use pessimistic locking?
Many server databases don’t offer row locking at all, they can only lock a whole memory page (usually between 2k and 64k of memory).  
The server databases that do offer row locking usually recommend to not to use it for efficiency reasons – it makes the database slower, and can lead to deadlocks.

The other reason that VFP does not even attempt to offer pessimistic locking against SQL tables is that a VFP remote view is NOT a direct connection to the database server.
The data in your remote view is a copy of the server data.
When you issue a TABLEUPDATE() command, VFP issues an update statement for each changed row.  The local copy of the server data is actually a dbf on your temp drive.  Locking and unlocking data rows on your local copy of the data cannot affect other users in any way.
My recommendation is to use Optimistic Table Buffering.  ( Mode = 5 )
As nice as it would be to lock edited records, if your users demand that you “upgrade” from VFP tables to a server database, you cannot use pessimistic.

Of course, your users don’t care about what VFP does or why it does it.
So, if you start with pessimistic buffering, they will demand the same functionality with SQL server – leaving you in a tough spot.

TableUpdate() and TableRevert()

TableUpdate() is the VFP function you use to save changes to a buffered table or view.
The really cool thing is that it works the same way against VFP tables, VFP views on VFP data, and VFP views on SQL data.  

Let’s use TableUpdate() on our form.

Open frmPlain.scx and add 2 buttons. 


Change the button properties as follows:

Command1.Caption = \<Save
Command1.Name = btnSave

Command2.Caption = Cancel
Command2.Name = btnCancel

Double click on the Save button and add code as follows:

* Procedure btnSave.Click()

#DEFINE ALL_ROWS 1

#DEFINE FORCE_UPDATE .T.

#DEFINE MB_OK 0

#DEFINE MB_INFO 64

=TABLEUPDATE(ALL_ROWS, FORCE_UPDATE, "person")

MESSAGEBOX("Record Saved",MB_OK + MB_INFO,"Save",800)
Double click on the Cancel button and add code as follows:

* PROCEDURE btnCancel.Click()

#DEFINE REVERT_ALL_ROWS .T.

=TABLEREVERT(REVERT_ALL_ROWS,"person")

ThisForm.Refresh()


Please note that table buffering works on one table at a time.
You could have some tables in buffering mode and other tables not in buffering mode.

Issuing TableUpdate() needs to be done for each table you have buffered.

Transactions

What is a transaction, and why do I need one?
A transaction makes your data changes an all or nothing deal.

The idea is that if something goes wrong, you do not end up with part of your changes saved.

Imagine the nightly processing at your bank.
Even if the power goes out, you really don’t want to remove money from Peter’s account if you don’t succeed in putting the money into Paul’s account.

Transactions let you do all your data changes to multiple tables in whatever order you want.  

You start a VFP transaction with BEGIN TRANSACTION
If every part of your data changes works ok, you END TRANSACTION to commit your changes.
If something goes wrong, you can explicitly ROLLBACK the changes.
If your computer dies, the power goes out, or whatever causes you to not END TRANSACTION to commit the transaction, then nothing happens.

Beginning and ending VFP Transactions does nothing to make a transaction against SQL Server or other server databases – you must send commands to the database server to make that happen.

VFP transactions are held in memory – in the event of a power failure or machine crash during your transaction, your changes are lost.

SQL Server and most other server databases log transactions to a log file on the server.
In the event that the server starts up and sees that there are unfinished transactions pending, it rolls back all changes to get back to the pre-transaction state.

Here is the save / cancel code with transaction support:

***************************************************

* Procedure btnSave.Click()

#DEFINE ALL_ROWS 1

#DEFINE FORCE_UPDATE .T.

#DEFINE MB_OK 0

#DEFINE MB_STOPSIGN 16

#DEFINE MB_INFO 64

BEGIN TRANSACTION

IF TABLEUPDATE(ALL_ROWS, FORCE_UPDATE, "person")


END TRANSACTION 


MESSAGEBOX("Record Saved",MB_OK + MB_INFO,"Save",800)

ELSE

ROLLBACK


MESSAGEBOX("Edits Lost!"+CHR(13)+"So Sorry!",MB_OK ;

+ MB_STOPSIGN,"Save",800)

ENDIF
***************************************************

The cancel does not need transactions – nothing ever got sent to the database. 













